Example of Z-transform

```
> restart:
```

Computation of Z transform of n^3 using different tecniques

USING MAPLE...

```
> ztrans( n^3, n, z) ;
```

$$\frac{z\left(z^2+4z+1\right)}{(z-1)^4} \tag{1}$$

USING the rule n^k f_n --> (-z*d/dz)^k f(z)

```
> HZ := z/(z-1) ;
```

$$HZ := \frac{z}{z-1} \tag{2}$$

```
> TMP := simplify(-z*diff(HZ,z)) ; # apply z*d/dz
```

$$TMP := \frac{z}{(z-1)^2} \tag{3}$$

```
> TMP1 := simplify(-z*diff(TMP,z)) ; # apply AGAIN z*d/dz
```

$$TMP1 := \frac{z\,(z+1)}{(z-1)^3} \tag{4}$$

```
> TMP2 := simplify(-z*diff(TMP1,z)) ; # apply AGAIN z*d/dz
```

$$TMP2 := \frac{z\left(z^2+4z+1\right)}{(z-1)^4} \tag{5}$$

USING the rule binomial(n,k) --> z/(z-1)^(k+1)

```
> B0 := 1 ;                  # binomial(n,0)
  B1 := n ;                  # binomial(n,1)
  B2 := n*(n-1)/2 ;          # binomial(n,2)
  B3 := n*(n-1)*(n-2)/6 ;    # binomial(n,3)
```

$$B0 := 1$$

$$B1 := n$$

$$B2 := \frac{1}{2}\,n\,(n-1)$$

$$B3 := \frac{1}{6}\,n\,(n-1)\,(n-2) \tag{6}$$

```
> COMBINE := collect(expand(x0*B0+x1*B1+x2*B2+x3*B3),n) ;
```

$$COMBINE := \frac{1}{6}\,n^3\,x3 + \left(-\frac{1}{2}\,x3 + \frac{1}{2}\,x2\right)n^2 + \left(x1 + \frac{1}{3}\,x3 - \frac{1}{2}\,x2\right)n + x0 \tag{7}$$

x3 and x0 are determined immediately

```
> SUBS := x3=6,x0=0;
```

$$SUBS := x3 = 6,\ x0 = 0 \tag{8}$$

```
> COMBINE := subs(SUBS,COMBINE) ;
```

$$COMBINE := n^3 + \left(-3 + \frac{1}{2}\,x2\right)n^2 + \left(x1 + 2 - \frac{1}{2}\,x2\right)n \tag{9}$$

x2 is determined

```
> SUBS := SUBS, op(solve( -3+(1/2)*x2, {x2} )) ;
```

$$SUBS := x3 = 6,\ x0 = 0,\ x2 = 6 \tag{10}$$

```
> COMBINE := subs(SUBS,COMBINE) ;
```

$$COMBINE := n^3 + (x1 - 1)\, n \tag{11}$$

x1 is finally determined

```
> SUBS := SUBS, x1=1 ;
```

$$SUBS := x3 = 6, x0 = 0, x2 = 6, x1 = 1 \tag{12}$$

Combine the Z transform of binomial coefficients

```
> simplify(subs( SUBS, x0*z/(z-1)+x1*z/(z-1)^2+x2*z/(z-1)^3+x3*z/
  (z-1)^4)) ;
```

$$\frac{z\left(z^2 + 4z + 1\right)}{(z-1)^4} \tag{13}$$

Compute Z-transform of 1+n+n^4

USING MAPLE...

```
> simplify(ztrans( 1+n+n^4, n, z)) ;
```

$$\frac{z\left(z^4 - 2z^3 + 14z^2 + 10z + 1\right)}{(z-1)^5} \tag{14}$$

OR USING the rule n^k f_n --> (-z*d/dz)^k f(z)

```
> ONE := z/(z-1);
```

$$ONE := \frac{z}{z-1} \tag{15}$$

```
> N := simplify(-z*diff(ONE,z)) ;
```

$$N := \frac{z}{(z-1)^2} \tag{16}$$

```
> N2 := simplify(-z*diff(N,z)) ;
```

$$N2 := \frac{z(z+1)}{(z-1)^3} \tag{17}$$

```
> N3 := simplify(-z*diff(N2,z)) ;
```

$$N3 := \frac{z\left(z^2 + 4z + 1\right)}{(z-1)^4} \tag{18}$$

```
> N4 := simplify(-z*diff(N3,z)) ;
```

$$N4 := \frac{z\left(z^3 + 11z^2 + 11z + 1\right)}{(z-1)^5} \tag{19}$$

```
> simplify(ONE+N+N4) ;
```

$$\frac{z\left(z^4 - 2z^3 + 14z^2 + 10z + 1\right)}{(z-1)^5} \tag{20}$$